W1DBM spent his summer designing and testing two-band vertical antennas via his computer. This summer we can enjoy the fruits of that labor plus get some time in on our computers.

Designing A Two-Band Loaded Vertical Antenna

BY PHILIP S. RAND*, W1DBM

I have used a two-band vertical phased array quite successfully for the past few years. This antenna consists of four 20 meter verticals with a 40 meter mobile resonator mounted on the top of each element. The idea occurred to me that it should be possible to redesign this antenna for use on other frequencies and for other purposes such as mobile whips, travel-trailer antennas, or low-frequency DX verticals.

After considerable researching of engineering handbooks and back issues of amateur magazines, in the September 1974 issue of *QST* I found an article entitled "Off-Center-Loaded Dipole Antennas" by Jerry Hall, K1PLP. A ¼-wave vertical, of course, is one half of a dipole. Therefore, all that would be necessary is to cut the overall length in half and wind only one loading coil. It was worth a try. However, the gigantic mathematical equation in this article (fig. 1) almost frightened me away. Since I had a computer, though, I figured if I could get the formula into a workable program I would be home free.

$$L_{\mu H} = \frac{10^6}{68\pi^2 f^2}\left\{\frac{\left[\ln\frac{24\left(\frac{234}{f}-B\right)}{D}-1\right]\left[\left(1-\frac{fB}{234}\right)^2-1\right]}{\frac{234}{f}-B} - \frac{\left[\ln\frac{24\left(\frac{A}{2}-B\right)}{D}-1\right]\left[\left(\frac{\frac{fA}{2}-fB}{234}\right)^2-1\right]}{\frac{A}{2}-B}\right\}$$

where

$L_{\mu H}$ = inductance required for resonance
ln = natural log
f = frequency, megahertz
A = overall antenna length, feet
B = distance from center to each loading coil, feet
D = diameter of radiator, inches

Fig. 1– The program formula from Jerry Hall's article "Off-Center-Loaded Dipole Antennas," which appeared in QST, *September 1974.*

Another article was found in *CQ* magazine, December 1981 by Dick Sander, K5QY, entitled "A Computer-Designed Loaded Dipole Antenna." This article lists a program for an Apple II computer that puts Jerry Hall's formula into a workable form, and some useful ideas were obtained from it.

Jerry's article describes the design of a dipole that is considerably shorter than normal (you select the length that you have available). It then calculates the necessary inductance for a pair of loading coils to make it resonate at the desired frequency. You must select the distance, "B," from the center insulator to each loading coil. Considering one half of the dipole, if you choose "B" as ¼ wave at a higher frequency, you will have a two-band, ¼-wave vertical antenna.

Fig. 2 shows a diagram of the loaded vertical. Note that the overall height, "H," resonates at a low frequency, while the length below the loading coil resonates at a higher frequency. This design requires that the overall height be less than ¼ wavelength at the lower frequency, and that the part below the loading coil be ¼ wavelength at the higher frequency. The overall length, of course, must be longer than a ¼ wave at the higher frequency. The loading coil must have enough inductance so that it will act like an RFC on the higher frequency and isolate the lower part of the antenna, while acting like a loading coil to resonate the entire antenna at the lower frequency. A good radial ground system will be necessary under the antenna for best results.

The Equation

The equation in *QST* is rather cumbersome to handle in its present form (see fig. 1). It is much easier to enter into the computer if you first divide it algebraically into several logical sections, calculating each separately and then recombining them later. Also, two parts of the equation are each used twice, so they are also calculated separately. Their values have been assigned to the variables "DD" and "GG." "F" is the lower frequency. "F2" has been added as the higher frequency. "F2" is used to determine the length "B" in fig. 2 automatically (B = 246 * K1 / F2 where K1 = K + .04). "K" is calculated from the ratio of the length to the diameter of the total antenna, while the ".04" was derived experimentally to compensate for the inductance of the loading coil that replaces the capacity of an end insulator. "D" is the diameter of the radiating element in inches. The computer program has a wire table built in so that the uH of the loading coil will be correct for any diameter of radiating element. You may enter either the diameter of the antenna or the wire gauge. You can therefore use this program to design a two-band ground-plane antenna with a large-diameter radiator as well as a wire vertical hung from a tree or a wooden pole.

The *QST* article only calculated the inductance in uH of the loading coils. Of course, you must wind these coils (see fig. 3). It is necessary, therefore, to find out the number of turns, wire size, length, etc. To determine the physical size of a single-layer air-wound coil where "EQ" equals inductance, the following formula

* *P.O. Box 8, Haverhill, NH 03765*

is entered into the computer program: "Inductance equals the number of turns, squared, times the coil radius, squared, divided by 9 times the radius plus 10 times the coil length," where **N** = number of turns, **A** = coil radius in inches, **B** = coil length in inches, and **EQ** = inductance in uH.

Rearranging the formula to solve for the number of turns and putting it into computer format, the formula becomes:

"N=SQR(EQ*(9*A+10*B))/A"

where **SQR** = square root, * = multiplication, and / = division. Since B is unknown, we enter B = 5 inches as a starter, and then "iterate" the equation 12 times, each time entering a new value for "B" by recalculating "B" = number of turns divided by turns-per-inch (for the size wire that you have selected). The number of turns-per-inch = **"WW"** and therefore, "B = N/WW." Usually, by the seventh or eighth iteration we have an accurate figure for both "N" and "B" and can proceed to wind the coils.

Practical Designs

To prove out this method of designing two-band shortened verticals, I have constructed several such verticals and have measured both the inductance of the computer designed coils and the s.w.r. of the finished antennas. In general, the



Fig. 2– Diagram of a two-band loaded vertical antenna.



LEGEND:
A = Coil radius (D/2)
B = Coil length
D = Coil diameter
N = Number of turns
EQ = Inductance (μH)

Fig. 3– Loading-coil winding criteria.

coils measured very close to the design inductance. Any difference was attributable to the unknown dielectric of the coil form (plastic water pipe). The lowest s.w.r. was not quite on the design frequency. This was attributable to ground conditions and to the proximity of nearby objects. With a small amount of pruning, it was possible to get all antennas down to an s.w.r. of less than 1.3/1 at the design frequency. The bandwidth for the higher frequency verticals was normal. As expected, the bandwidth of the lower frequency verticals decreased in proportion to the amount of shortening.

Antenna No. 1

This is a two-band travel-trailer antenna designed for use on the 20 and 40 meter Recreational Vehicle Service Nets on 7234 and 14308 kHz. A height of 20 feet was chosen for the vertical, which was mounted on my Airstream's rear bumper. The following data was entered into the computer:

F = 7.2; F2 = 14.3; H = 20; D = .625 (average dia. of mast, 10 ft. each of ½ in. and ¾ in. electrical conduit).

The computer said we needed a coil of 35.3 uH and that it should be placed 17.72 ft. above the bottom. Next we entered the coil diameter in inches and the size of the wire we planned to use.

D = 1.0 in.; WW = 14-gauge enameled magnet wire.

The computer calculated the following data for the loading coil:

Coil length = 6.69 in.

No. of turns = 100.4 using No. 14 enameled wire close-wound. (An 8 in. length of 1 in. dia. plastic water pipe was used for the form).

Antenna No. 1 performed well. The bandwidth on 40 meters was quite a bit narrower than a full-size vertical due to its shorter length. The 20 meter bottom section resonated close to the design frequency and worked like any other 20 meter ground-plane.

Antenna No. 2

This two-bander was for 3.9 and 21.3 MHz with an overall height of 15 ft. and a diam. of 1.25 in. (TV masting). The computer calculated the bottom-to-coil distance as 11.9 ft. and the coil inductance as 92.9 uH. It was decided to use a 12 in. length of 1.5 in. dia. plastic pipe for the coil form. The coil length is 9.05 in. wound with 126.8 turns of No. 20 insulated hookup wire.

Antenna No. 2 had a much narrower bandwidth than a normal-size dipole on 75 meters due to its much shorter overall length. It did, however, have a better bandwidth than a mobile whip with only a 5 ft. bottom section, and it also got out much better. On 15 meters it appeared to function normally. Slight pruning was required to get the s.w.r. down to a low value, especially on 75. For solid-state rigs a "match-box" solves the s.w.r. problem when QSYing too far from the design frequency.



Fig. 4– A two-band, low-frequency DX antenna. See "Antenna No. 3" section for details.

Antenna No. 3

This is a two-band, low-frequency DX antenna. A wire version of this antenna has been tested in a horizontal position as a dipole and should work much better in the vertical position if a suitable ground-plane is provided.

Frequencies: 1.82 and 3.8 MHz
H = 70 ft. (total height)
D = 8 in. (tower, 8 in. to a side)
Coil dia. = 4 in. plastic drain pipe
Coil wire = No. 14 insulated house wire

The computer says: mount the loading coil 66.03 ft. from the bottom (see fig. 4). The coil should be 144 uH, 59 turns with a winding length of 7.9 in. The top section should be long enough to make the total height 70 ft. Adjust bottom length for lowest s.w.r. on 3.8 and then the top section for lowest s.w.r. on 1.82.

Computer Program

The program is made up of a bunch of numbered lines telling the computer in numerical sequence what you want it to do. Lines 10 through 200 ask you to input data such as frequencies, overall length, "want to use wire table?" etc. The actual computation starts at line 200. A "GOSUB 1280" statement sends the computer to line 1280, a sub-routine, to calculate the value of "K" from the length-to-diameter ratio. It then returns to line 210 to calculate the value of "B," the length of the higher frequency vertical.

Program Listing

```
10 CLS
20 PRINT" A PROGRAM FOR:"
30 PRINT" DESIGNING A 2-BAND SHORTENED"
40 PRINT" AND LOADED VERTICAL ANTENNA."
50 PRINT" WRITTEN BY PHIL RAND-W1DBM."
60 PRINT" JULY 15, 1983
70 PRINT" ENTER A FREQ. IN LOWER BAND:"
80 INPUT F
90 PRINT" ENTER A FREQ. IN HIGHER BAND:"
100 INPUT F2
110 PRINT" ENTER OVERALL VERT. ANT.HEIGHT"
120 INPUT H
130 A=2*H
140 PRINT" WANT TO USE ANT. WIRE TABLE? ANSWER 'Y' OR 'N':"
150 INPUT Y$
160 IF Y$="Y" THEN 1090
170 IF Y$="N" THEN 1240
180 IF Y$<>"Y" THEN 150
190 IF Y$<>"N"THEN 150
200 GOSUB 1280
210 B=246*K1/F2
220 SL=B
230 PI=3.1416
240 AA=10^6/(68*PI^2*F^2)
250 DD=234/F-B
260 GG=A/2-B
270 BB=LOG(24*DD/D)-1
280 CC=(1-F*B/234)^2-1
290 EE=LOG(24*GG/D)-1
300 FF=(F*GG/234)^2-1
310 EQ=AA*((BB*CC/DD)-(EE*FF/GG))
320 EQ=INT(EQ*100+.5)/100
330 PRINT
340 CLS
350 PRINT" INDUCTANCE OF THE LOADING COIL IS"EQ"UH."
360 PRINT"** CALCULATING SIZE OF COIL **"
370 PRINT" ENTER DIA. OF LOADING COIL:"
380 INPUT DL
390 A=DL/2
400 B=5
410 CLS
420 PRINT" ENTER THE WIRE GAUGE OF THE COIL, 12 TO 16."
430 REM-
440 PRINT"            - OR -"
450 PRINT" ENTER '1' FOR B&W 10 TURNS PER INCH COIL STOCK."
460 PRINT" ENTER '2' FOR N0.12 GA. INSULATED HOUSE WIRE."
470 PRINT" ENTER '4' FOR NO.14 GA. INSULATED HOUSE WIRE."
480 PRINT" ENTER '6' FOR NO.20 INSULATED HOOK-UP WIRE."
490 INPUT W
500 IF W=1 THEN WW=10
510 IF W=2 THEN WW=6.5
520 IF W=4 THEN WW=7.5
530 IF W=6 THEN WW=14
540 IF W=12 THEN WW=12
550 IF W=13 THEN WW=13.5
560 IF W=14 THEN WW=15
570 IF W=15 THEN WW=16.8
580 IF W=16 THEN WW=18.9
590 CLS
600 PRINT" < ITERATING FORMULA >"
610 PRINT TAB(7)"TURNS";TAB(15)"LENGTH"
620 X=1
630 FOR I=1 TO 12
640 N=SQR(EQ*(9*A+10*B))/A
650 N=INT(N*100+.5)/100
660 B=INT(B*100+.5)/100
670 PRINT"NO."X;TAB(8)N;TAB(15)B
680 B=N/WW
690 B=INT(B*100+.5)/100
700 X=X+1
710 NEXT I
720 PRINT" PRESS ENTER/RETURN"
730 INPUT C
740 CLS
750 PRINT" *** ANTENNA AND COIL DATA ***"
760 PRINT" DIA.OF ANT.ELEMENT IS"D"IN."
770 PRINT" FREQ.S ARE:"F"AND"F2"MHZ."
780 PRINT" HT.OF VERT.ANT.IS"H"FEET"
790 SL=INT(SL*100+.5)/100
800 PRINT" LENGTH FROM BOTTOM OF ANT."
810 PRINT" TO COIL IS"SL"FEET."
820 TL=H-SL
830 PRINT" LENGTH-TOP SECTION:"TL"FT."
840 PRINT" INDUCT. OF COIL IS"EQ"UH."
850 PRINT" DIA. OF COIL IS"DL"INCHES."
860 PRINT" LENGTH OF COIL IS"B"IN."
870 PRINT" NO. OF TURNS-PER-IN. IS";WW
880 PRINT" TOTAL NO. OF TURNS IS";N
890 IF W=1 THEN 960
900 IF W=2 THEN 990
910 IF W=4 THEN 1020
920 IF W=6 THEN 1050
930 PRINT " USE"W" GAUGE ENAMMELED MAGNET"
940 PRINT" WIRE FOR WINDIND COIL."
950 GOTO 1080
960 PRINT" USE B&W 10 TURNS-PER-INCH COIL"
970 PRINT" STOCK FOR THE COIL."
980 GOTO 1080
990 PRINT" USE NO.12 INSULATED HOUSE WIRE"
1000 PRINT" FOR THE COIL."
1010 GOTO 1080
1020 PRINT" USE NO.14 INSULATED HOUSE WIRE"
1030 PRINT" FOR THE COIL."
1040 GOTO1080
1050 PRINT" USE NO.20 INSULATED HOOK-UP"
1060 PRINT" WIRE FOR THE COIL."
1070 GOTO 1080
1080 END
1090 PRINT" ENTER ANT. WIRE GA., 10 TO 16,"
1100 PRINT" IN EVEN NUMBERS."
1110 INPUT GA
1120 IF GA=10 THEN D=.1019
1130 IF GA<>10 THEN 1150
1140 IF D=.1019 THEN 200
1150 IF GA=12 THEN D=.0808
1160 IF GA<>12 THEN 1180
1170 IF D=.0808 THEN 200
1180 IF GA=14 THEN D=.064
1190 IF GA<>14 THEN 1210
1200 IF D=.064 THEN 200
1210 IF GA=16 THEN D=.0508
1220 IF GA<>16 THEN 1240
1230 IF D=.0508 THEN 200
1240 PRINT" ENTER ANT. ELEMENT DIA. IN IN.:"
1250 INPUT D
1260 GOTO 200
1270 REM L=ANT. LENGTH IN FT.
1280 L=A
1290 DF=D/12
1300 R=L/DF
1310 IF R<5.0 THEN K=.90
1320 IF K=.90 THEN 1480
1330 IF R<8  THEN K=.91
1340 IF K=.91 THEN 1480
1350 IF R<13 THEN K=.92
1360 IF K=.92 THEN 1480
1370 IF R<17 THEN K=.93
1380 IF K=.93 THEN 1480
1390 IF R<25 THEN K=.94
1400 IF K=.94 THEN 1480
1410 IF R<50 THEN K=.95
1420 IF K=.95 THEN 1480
1430 IF R<250 THEN K=.96
1440 IF K=.96 THEN 1480
1450 IF R<5000 THEN K=.97
1460 IF K=.97 THEN 1480
1470 IF R=>5000 THEN K=.98
1480 K1=K+.04
1490 RETURN:REM (TO 390)
```

Next it starts the calculation of the loading-coil inductance by defining the values of the numeric variables: B = 246 * K1 / F2; SL = B; PI = 3.1416; DD = 234 / F − B; GG = A / 2 − B. The actual equation for the inductance starts at line 240 and has been broken down into:

EQ = AA*((BB*CC/DD) − (EE*FF/GG))

where EQ is in uH and AA, BB, CC, DD, EE, FF, and GG are parts of the original equation (fig. 1).

Designing The Coil. Lines 420 to 560 give you your choice of a number of different kinds of wire for winding the coil. Lines 580 to 710 give the computer instructions for "ITERATING" the no.-of-turns equation:

N = SQR(EQ*(9*A + 10*B))/A

Computer Printout. Lines 730 through 1060 tell the computer to display the results on the TV screen. Lines 1100 to 1260 contain the antenna wire table.

Program Listing. In the accompanying program listing, whenever you see a line containing "CLS", (command to "CLEAR THE SCREEN"), you must change it to one of the following, depending upon your brand of computer. For example, if you own an:

Apple II—change "CLS" to "HOME"

Vic-20—change "CLS" to "PRINT "CLR/HOME" "

TI-99/4A—change "CLS" to "CALL CLEAR"

For the Timex-Sinclair, IBM, and TRS-80—"CLS" is okay.

This complete program requires only about 3.5K of usable RAM and the use of LOGs. (Extended BASIC required for the TRS-80 Color Computer and the TI99/4A.) It may be run on a Vic-20 with 3583 BYTES FREE. You will need extra memory for the Timex/Sinclair, which also requires the addition of the word "LET" when assigning values to numeric variables. (For example: 210 B = 246*K1/F2, becomes 210 LET B = 246*K1/F2, etc.)